Stellar Structure with Maple (sll --- v1.0, January 2012)

The Basics of Stellar Structure

Stars are complicated objects, as we have seen. The physics underpinning the interior structure of stars is straight-forward enough, but the mathematics describing that physics is a mess! There are four basic equations for the parameters of interest, all radially dependent: the pressure $P(r)$, the mass $M(r)$, the luminosity $L(r)$ and the temperature $T(r)$. The differential equations describing these four variables are summarized in pg. 330 of BOB, where they are still coupled, but have all be expressed in terms of a single unknown radially dependent physical quantity: the density $\rho(r)$.

Note that the pressure $P(r)$ and mass $M(r)$ characterize the mechanical structure of the star, whereas the luminosity $L(r)$ and temperature $T(r)$ characterize the energy. It was realized early on that if one knew how to relate pressure P to density ρ then the mechanical equations could be solved independently of the energy equations.

Generic relationships between P and ρ do not exist (otherwise we wouldn't have a big set of complicated stellar structure equations!), but certain theoretical models suitable for a wide range of assumptions about the stars have great practical utility. A widely used class of models are ***polytropes***, which will be the subject of this exercise.

Polytropes

A *polytrope* is a sphere of gas with the pressure P related to the density ρ by[1]

$$P = \kappa\rho^{\gamma} \qquad \rightarrow \qquad \gamma \equiv \frac{n+1}{n} \qquad \rightarrow \qquad P = \kappa\rho^{(n+1)/n}$$

where κ is a constant throughout the star (but could be different for different stars), and n is called the *polytropic index*. This kind of relationship is called an *equation of state*. Polytropes are very useful in all kinds of stellar modeling, depending on the value of the index γ.

The κ constant ▶ ....

What is this κ constant? Remember that κ is constant throughout the star, so imagine looking at the equation of state at the core where $P = P_c$ and $\rho = \rho_c$. Now look at P and ρ at any other radius r in the star. I can write the ratio of the equation of states as (note the κ cancel out)

$$\frac{P(r)}{P_c} = \left(\frac{\rho(r)}{\rho_c}\right)^{(n+1)/n} \qquad \rightarrow \qquad P(r) = \left(\frac{P_c}{\rho_c^{(n+1)/n}}\right)\rho(r)^{(n+1)/n}$$

Comparing this to the polytrope equation, we can now note that physically κ characterizes the central pressure and central density in our model.

The Lane-Emden Equation

If you make the choice of a polytropic equation of state, then the mechanical equations of stellar structure can be combined to form a single, master equation for the structure of our star. We will

[1]Remember that $P = P(r)$ and $\rho = \rho(r)$; I have suppressed the (r) here for clarity.

not pursue that derivation here; it is outlined in great detail in §10.5 (pg. 335-336) of BOB. The result is the *Lane-Emden Equation*.

The Lane-Emden Equation is a second-order differential equation (up to 2 space derivatives) for the *spatial structure* of the star. For reasons we discuss below, it is written in terms of two *normalized* variables[2]: x and Y.

The radial coordinate naturally gets scaled by a recurring quantity r_n with dimensions of length, so we define

$$x = \frac{r}{r_n} \qquad \rightarrow \qquad r_n = \sqrt{\frac{(n+1)P_c}{4\pi \, G \, \rho_c^2}}$$

At this point you can choose to work either in terms of the pressure or the density; it doesn't matter which you choose, since they are related by the polytropic equation of state. We will follow BOB, and work with ρ. We define $Y(r)$ as

$$Y(r) = \left(\frac{\rho(r)}{\rho_c}\right)^{1/n}$$

Note that I have dropped the rather cumbersome notation that BOB uses where there is a subscript n being carried around; this is notational only and has no mathematical function.

Using these variables, the *Lane-Emden Equation* may be written as

$$\frac{d^2Y}{dx^2} + \frac{2}{x}\frac{dY}{dx} + Y^n = 0$$

As it turns out, this equation can be solved analytically for $n = 1$ and $n = 2$, but not for $n = 3$. This is our target equation to solve in `Maple`. The $n = 3$ model is a good approximate for stars like the Sun, so let's use `Maple` to make a model of the Sun.

[2]Note BOB uses D for Y; θ is also common in other books. I find both D and θ confusing because I use them for other things. I don't use Y for *anything*!

Setup in Maple

`Maple` ultimately is going to make plots for us, so it has to have numerical values for everything, in particular for constants that have physical meaning.

Variable Assignments ▶
You can make variables of any name you want in `Maple` and store numbers in them. The creation and storage of a number can be handled all at once using the assignment operator, `:=`

Let's enter all the constants for this problem in SI units:

Maple Entry	***Comment***
`[> n := 3;`	Polytropic index
`[> G := 6.67384e-11;`	Newton's constant
`[> Rsun := 6.955e8;`	Solar radius (m)
`[> rhoc := 160000;`	Density at core (kg/m^3)
`[> Pc := 10^(16.53);`	Pressure at core (pascals)
`[> kappa := Pc/rhoc^((n+1)/n);`	Equation of state constant κ
`[> rn := evalf(sqrt((n+1)*Pc/(4*Pi*G*rhoc^2)));`	Scaling length

The Lane-Emden Equation ▶
Next, we have to teach `Maple` the Lane-Emden equation. I will store the entire equation in a variable assignment to make it easier to use it later on. Here we use the `Maple` command `diff` where `diff(Y(x),x$n)` $= d^nY(x)/dx^n$.

```
[> leEqn := diff(Y(x),x$2) + (2/x)*diff(Y(x),x) + Y(x)^n = 0;
```

Numerical Solution in Maple

As we have said, $n = 3$ cannot be solve analytically. Please try now, I'll wait. ♪♪♪ See?

We will use Maple to solve it numerically. Maple has a built-in numerical differential equation solver called dsolve(), which takes as its input the differential equation of interest, some information about the domain you want a solution on, and the boundary conditions on the equation and its derivatives.

Boundary Conditions ▶ . . . .
To solve a second order differential equation, you have to give initial conditions on the the dependent variable (Y in this case) and its first derivative. The initial conditions are boundary conditions, and we are starting our model with known conditions at the core of the star, $x = 0$. Using our identity for $Y(r)$ we see that at $r = 0$ we have $Y(0) = 1$. Furthermore, we don't want anything pathological to happen to the model at the core, so we insist that the first derivative go to zero as well: $dY/dx|_{x=0} = 0$.

```
[> leIC := Y(0) = 1, D(Y)(0) = 0;
```

The ODE Package ▶ . . . .
I find it convenient to keep the equation and initial conditions together for input into dsolve(), so

```
[> ODE := {leEqn,leIC};
```

Now all there is to do is solve it! We want to have a numeric solution for $Y(x)$, so

```
[> leSoln := dsolve(ODE,Y(x),type=numeric,output=listprocedure);
```

The output=listprocedure option makes the output useable as a ordinary Maple style function in our analysis that follows.

Maple's Solution ▶ . . . .
At this point, Maple has stored its solution in the variable leSoln. It is a function of x, so if you give it a value of x, it will spit an answer back at you! Try a value:

```
[> leSoln(0.8);
```

You get a list of three values back! The value of x, the value of $Y(x)$ and the value of dY/dx. For our purposes here, what we really want is just the second bit — the solution $Y(x)$, as that is what our physical functions are built on.

```
[> Ysoln := x -> rhs(leSoln(x)[2]);
```

Let's parse what that command does. This creates a new function Ysoln(x) that I can call. It makes this function equal to whatever Maple is using to get the right-hand-side (rhs) of the thing that is the second slot of leSoln.

At this point, Ysoln(x) is a solution to the Lane-Emden equation, and can be used like any other function in Maple. Try plotting it:

```
[> plot(Ysoln,0..150);
```

Our job now is to turn this into physics. `:-)`

Getting the Physics Out: Density ρ ▶ ....
We can get the density from the definition of $Y(r) = Y(x \cdot r_n)$.

```
[> rho := x -> rhoc*Ysoln(x)^n;
```

Now ρ`(x)` is a solution for the density as a function of x.

Getting the Physics Out: Pressure P ▶ ....
With density ρ, we can get pressure from the equation of state.

```
[> P := x -> kappa*rhoc^((n+1)/n) * Ysoln(x)^(n+1);
```

Now `P(x)` is a solution for the pressure as a function of x.

Getting the Physics Out: Stellar Surface ▶ ....
This solution is only good out to the surface of the star. We can find the surface in the numerical solution by looking for the place where the density goes to zero. That can be accomplished by using the `Maple` command `fsolve()`, which takes as input the function we want to find the zeroes of and the range over which to find zeros. The easiest way to figure out the range to search over is to make a plot.

```
[> plot(rho,0..50);
```

```
[> Xsurf := fsolve(Ysoln(x)=0,x=0..10);
```

Making some plots ▶ ....
We now have in our hands everything we need to look at the structure of the star, as given by the polytrope model. Fundamentally, what we care about as physicists are plots of the density ρ and pressure P as a function of radius. We can do this in terms of the dimensionless variable x:

```
[> plot(rho,0..Xsurf);
```

```
[> plot(P,0..Xsurf);
```

Unfortunately, that is kind of annoying because you probably don't have a good built in intuition about how x relates to the radius r (at least not so if I tell you x you can instantly tell me what the corresponding value of r is!). So let's plot as a fraction of the stellar radius: $R = r/R_\star$. Remember that `Maple` thinks of `rho = rho(x)` and `P = P(x)` so let's exploit their function like nature. Consider: `R Xsurf` is a quantity that corresponds to $x = 0$ when $R = 0$, and $x = X_{surf}$ when $R = 1$. So:

```
[> plot(rho(R*Xsurf),R=0..1);
```

```
[> plot(P(R*Xsurf),R=0..1);
```

These are now plots of ρ and P as functions of the stellar radius. *Mischief managed.*

Some References

- ▷ Far and away one of the most widely recognized monographs on the topic of stellar structure is *An Introduction to the Study of Stellar Structure* by Chandrasekhar. Chandra was famous for working on a topic obsessively for a decade or so then writing a definitive books summarizing everything he found out. This is the one for stellar structure.
- ▷ BOB talks about the Lane-Emden equation on pgs. 334-340. A reasonably sophisticated stellar structure code, `StatStar`, is detailed in Appendix L. The code (in FORTRAN and C++ versions) may be downloaded from the book website at `<www.aw-bc.com/astrophysics>`. Click on *Online Resources*.
- ▷ An excellent book on computational astrophysics is *Astrophysics with a PC* by Paul Hellings. The book is easy to read, with small introductory bits about the physics, followed by small introductory bits about numerical methods, and then complete codes (written in `BASIC`) that solve the astrophysics problems that have been described. Chapter 7 and 8 are about stellar structure, Chapter 9 is about stellar atmospheres, and Chapter 10 is about the structure of white dwarfs. The book is available directly from Willmann-Bell (the publisher) for $20. `<http://www.willbell.com/math/mc5.htm>`